
ABSTRACT  

‘NOTE:  No unequal values were found.  All values compared are exactly equal.’  Are you excited to see 
these statements in your SAS® output?   After laboring for hours, finally, you got a match!  Eureka!  You 
finally did it!  But before you pop a champagne bottle and celebrate, please note these statements are 
necessary but not sufficient for a perfect match!   In this paper, the author explains why the 
aforementioned statements may not identify a perfect match.  She introduces the use of &SYSINFO, an 
automatic macro variable generated by PROC COMPARE.  This single value held by &SYSINFO enables 
users to know accurately the result of the comparison, which includes a perfect match. After a brief 
introduction, the author demonstrates how users can take advantage of the features of &SYSINFO to 
speed up the validation of numerous deliverables such as datasets, tables, listings and figures, and how 
users can use traffic-lighting technique and Excel spreadsheet features to make the result more readable. 
Armed with these techniques, speeding up the validation process is really as easy as 1, 2 and 3. 

INTRODUCTION  

To ensure the accuracy of a Clinical Study Report (CSR), analysis datasets, tables, listings and figures 
generated by the production programmer are often validated by a validation programmer.  If SAS is the 
language of choice, the COMPARE procedure offers nice features to facilitate the validation process. The 
automatic macro variable &SYSINFO generated by the COMPARE procedure enables users to quickly 
access the result of the validation and avoid the pitfalls of misinterpreting the comparison results as 
stated in Horstman and Muller (2013).   

Below are the topics that will be covered in this paper. 

• Don’t get blindsided by PROC COMPARE  

• An Introduction to &SYSINFO 

• Interpretation of the value of &SYSINFO  

• Expand the use of &SYSINFO.  Generate a SAS dataset that summarizes the validation result of 
multiple deliverables, such as all tables in a CSR. 

• Use traffic-lighting techniques to alert users of any serious mismatches 

• Put validation results in an Excel Spreadsheet to make the results more readable. 

OVERVIEW  

PROC COMPARE is a very powerful SAS® procedure that enables users to compare two datasets, to 
compare variables against variables of the same dataset or variables against variables between two 
datasets.  It is relatively simple, yet powerful procedure to use once users have mastered its features.  
According to SAS® Procedure Guide 9.4, the basic syntax for COMPARE procedure is as follows: 

 

PROC COMPARE <option(s)>; 

    BY <DESCENDING> variable-1 

       <DESCENDING> variable-2 …  <NOTSORTED>; 

    ID <DESCENDING> variable-1  

       <DESCENDING> variable-2 … <NOTSORTED>;  

     VAR variable(s); 

       WITH variable(s); run; 

Program 1. Syntax for PROC COMPARE 

 
PROC COMPARE has numerous options that enable users to list all variables, all observations, control 
the display of your comparison output, generate output datasets, suppress the writing of observations to 
output datasets when all values in the observations are judged to be equal, etc. Please familiarize with 
these options to make the most use of what PROC COMPARE has to offer. Usage of ID statement to 
uniquely identify the observations for comparison is highly recommended.  VAR statement enables one to 
select the variables to be compared.  This statement can be useful when we want to concentrate on 
certain variable(s) in comparison.  Since we are mainly interested in comparison of 2 datasets, the WITH 
statements are beyond the scope of this paper. 
 
For more information on PROC COMPARE, please refer to Base SAS® 9.4 Procedures Guide, Seventh 
Edition, Pages 419-488. 

DON’T GET BLINDSIDED BY PROC COMPARE 

The ultimate goal for a validation programmer is to identify and resolve any discrepancies to get a perfect 
match!    So imagine how excited a validation programmer would be to see the following statement in the 
PROC COMPARE output. 

 
NOTE: No unequal values were found. All values compared are exactly equal.   

 

Eureka!  It is a match!  Well, these statements are necessary but not sufficient for a perfect match!  Have 
you checked if the same variables are in both datasets?  Are all variables of the same type?  Do both 
datasets contain the same observations?  The truth is you can have discrepancies in your datasets and 
still get the forementioned statements in your PROC COMPARE output!  As Horstman and Muller (2014) 
have warned us, ‘Don’t get Blindsided by PROC COMPARE’. 
 
Consider the following 2 datasets:   
  

        
 

Example 1.  CLASS_PROD vs CLASS_QC 
 

Just at first glance, you can see how different CLASS_PROD and CLASS_QC are! 
 

1. CLASS_PROD has 3 variables and CLASS_QC has 4 variables.   
       SCORE exists only in CLASS_QC. 
2. CLASS_PROD has 4 observations and CLASS_QC has 6 observations. 
3. SEX is a character variable in CLASS_PROD and a numeric variable in CLASS_QC. Let us 

compare these 2 datasets with PROC COMPARE.  

proc compare base=CLASS_PROD compare=CLASS_QC LISTVAR LISTOBS; 

   id SUBJID; 

   title ‘CLASS_PROD vs CLASS_QC’; run; 

 

     Program 2. Use PROC COMPARE to compare CLASS_PROD vs CLASS_QC 

(Note that LISTVAR option lists all variables found in only one 

dataset and LISTOBS option lists all observations found in only one 

dataset.) 
 

Below is the result from PROC COMPARE 
 

CLASS_PROD vs CLASS_QC 

As illustrated by this example, ‘NOTE:  No unequal values were found.  All values compared are exactly 
equal.’  are not reliable statements to identify a perfect match!  So how can we identify a perfect match 
then?  &SYSINFO comes to your rescue!    

 
(Note:  the author does not think these statements are erroneous, even though they can be misleading!  
SAS clearly states that ‘All values compared are exactly equal.’  So if there are extra variables, mismatch 
in types, mismatching in ID variables or extra observations without matching ID variables, SAS cannot do 
the comparison and hence, not compared.   SAS only claimed that the variables SAS could compare 
were exactly equal; it did not say that the BASE dataset and the COMPARE dataset were perfectly 
matched!) 

WHAT IS &SYSINFO? 

&SYSINFO is an automatic SAS macro variable generated by PROC COMPARE to store the return code.  
The value of the return code provides information about the result of the comparison.  Table 1 displays 
the 16 possible discrepancy conditions.  Each condition has a numeric value, displayed in Base 10 and 
binary 16. format.   &SYSINFO  is the sum of the values for any of the 16 conditions violated.  Please get 
the return code values immediately after PROC COMPARE, before another procedure or data step. You 
may want to store the value in another macro variable such as &RC because its value will be reset to 
zero once another procedure or data step begins. 

 
proc compare base=base data compare=compare data;  

run; 

%let RC=&SYSINFO; 

%put &=RC;  

Program 3. PROC COMPARE generates &SYSINFO, whose value is stored in &RC. 

Table 1.  The Return Code Values  

HOW TO INTERPRET THE VALUE OF &SYSINFO 

&SYSINFO holds the sum of the coded values of any 16 conditions in Table 1, if the corresponding 
condition has been violated.  Each value represents a unique combination of the 16 conditions.  Suppose 
you run the code like those in Program 3 and get the value 48.  What does that mean?  Which of the 16 
conditions are violated?  Let us put the number 48 in binary16. format.   

 

 

The 5th and 6th positions from the right have a value of 1 meaning that the 5th and 6th conditions stated in 
Table 1 have been violated.  In other words, there is at least one variables of different length and at least 
one variables of different labels between the base dataset and compare dataset. 

The design of &SYSINFO is based on the binary data representation method which was first introduced in 
late 1880s where the German-American engineer, statistician and inventor Herman Hollerith first made 
use of punch cards to complete 1890 US census.  Nevertheless, it is truly ingenious!  Putting its value in 
binary format, one can see that each of the16 conditions is represented by one of the 16 positions!  If the 
condition is true, its value in the corresponding position is a 1; otherwise, a zero.  That is why each value 
of &SYSINFO represents a unique combination of violations!  And one single number tells you the result 
of your comparison!   

And the ingenuity of &SYSINFO does not stop here!  What is the commonality of the first 6 conditions? 
These conditions are attributes of datasets or variables!  Since the maximum value one can get by adding 
up all the 6 conditions is 63. Any &SYSINFO from 1 to < 64 means difference in attributes only.  In some 
cases, difference in attributes may not be that critical!  What about &SYSINFO=0?  That means none of 
the 16 conditions have been violated!  That is a perfect match!  So in a nutshell, the high-level 
interpretation of &SYSINFO is: 

 

       &SYSINFO = 0      means       Perfect Match! 

       &SYSINFO < 64    means       Difference in Attributes only. 

       &SYSINFO >=64   means       Some Serious Discrepancies! 

 

Table 2.  High Level Interpretation of the value from &SYSINFO. 
 

Is a ‘Perfect Match’ required?  The answer is probably Yes for analysis dataset validation.  However, 
when validating the dataset used for PROC REPORT (One-Proc-Before dataset), difference in 
attributes may not be that critical.  In that case, a &SYSINF0 < 64 may suffice.  So it depends what 
you are validating. 

Armed with this knowledge, let us re-consider Example 1.   
 

 
 
The following discrepancies have been observed. 
 
1. CLASS_PROD has 3 variables and CLASS_QC has 4 variables.   
       SCORE exists only in CLASS_QC.   
      (That is, Condition 8 in Table 1, having a value of 128.)  
2. CLASS_PROD has 4 observations and CLASS_QC has 6 observations. 

(That is, Condition 12 in Table 1, having a value of 2048.)  
3. SEX is a character variable in CLASS_PROD and a numeric variable in CLASS_QC. Let us 

compare these 2 datasets with PROC COMPARE. 
(That is, Condition 14 in Table 1, having a value of 8192.) 
 

Since &SYSINFO is the sum of all violation conditions, the expected value of &SYSINFO will be: 
                                        
                                        128 + 2048 + 8192 = 10368.   
 
Let us double check the result!   
 
 

proc compare base=CLASS_PROD compare=CLASS_QC; 

   id SUBJID; 

   title ‘CLASS_PROD vs CLASS_QC’; 

run; 

%let RC=&SYSINFO; 

%put &=RC; 

 
 

Program 4. Get the &SYSINFO value for CLASS_PROD and CLASS_QC comparison. 
 
(Note: The value of &SYSINFO can be directly displayed here without using &RC.  However, it may be a good habit 
to start saving the &SYSINFO value since its value will be reset to zero once another procedure or data step begins.) 

Program 4 Output:  RC=10368 is as expected.  It is the total sum of the values for the 3 violation 
conditions. 
 
 
One may wonder that given RC=10368, can we easily figure out which conditions have been violated?  Although it is 
not that intuitive, the violation conditions can be easily identified by Bit Map Testing.  Bit Map Testing checks for the 
value of 1 in a position, and decides if the corresponding conditions have been violated.  Partial code for Bit Map 
Testing can be found in Base SAS®9.4 Procedure Guide, Seventh Edition, Page 448. 
 
 

%let RC=&SYSINFO; 

data _null_; 

  /* Test for data set label */ 

  if &rc = ‘1’b then 

     put ‘<<<< Data sets have different labels’; 

  /* Test for data set types */ 

  If &rc = ‘1.b’ then 

     put ‘<<< Data set types differs’; 

 

     ... more similar codes for the other conditions ...   

   run; 

 

 
Program 5: Code for Bit Map testing can be used to the violated condition(s).  
 
Based on the code above, the author developed the macro %INTERPRET that can easily decipher all violation 
conditions.  (Please see %INTERPRET in Appendix 1.) Moreover, result of each violation condition is stored in a 
macro variable.  For instance, if Condition 1 has been violated &RC_1 will be resolved to Y and similarly, for other 
conditions.   
 
 
 
 



 

9 

 

So if at the end of Program 4, we add: 
 
 

     %INTERPRET (RC=&RC);  

Program 4 (continue).  Invocation of %INTERPRET 
 

 

Program 4 (continue) Output.  Invocation of %INTERCEPT result in the SAS Log. 
 
Based on the SAS log above, one can see that for a RC of 10368, Conditions 8, 12 and 14 have been violated. 
Moreover, the corresponding macro variables for these conditions &RC_8, &RC_12 and &RC_14 all have a value of 
Y to indicate violation.   
 
Before leaving this section, the author would like to point out that one can also decipher the value of &SYSINFO by 
means of a function named bAND or bitwise AND.  Interested readers should review the paper by Hinson and 
Coughlin (2012). 
 

CAN THERE BE MORE THAN 16 VIOLATION CONDITIONS 
 
Many years ago, when the author first encountered the &SYSINFO, there were these 16 conditions.  Now  with 
ARTEMIS II going to the moon, Internet, Social Media, AI, SAS Version 9.6m9 and SAS/Viya, we still have these16 
conditions!   Should more conditions be added? 
 
Scrutinizing over the 16 conditions, the author noted that Condition #15 is ‘BY variables do not match’!  What about a 
condition for ‘ID variables do not match’ ?  The world is getting smaller now; sometimes, we get data in different 
languages.  What about another condition on ‘Difference in Encoding’?  We have encoding displayed in PROC 
CONTENTS output, why not consider its discrepancy in Encoding in PROC COMPARE?  (That said, the author 
would like to point out that FDA states in “Study Data Technical Conformance Guide, v6.1” Section 3.1.5  that “Use 
UTF-8 for extending character sets; however, the use of extended mappings is not recommended.”) 
 
If these 2 conditions are to be added, should those be Conditions #17 and #18?  Note that the current Condition #16 
is ‘Fatal error: comparison not done’.  That can happen if users attempt to compare two very different datasets, such 
as, comparing an Apple to an Orange!   I believe this Fatal Error is intentionally placed as the last condition, just like 
the first 6 conditions are the comparatively mild discrepancies, like difference in data/variable attributes.  So can we 
include these 2 perspective conditions to be Conditions #16 and #17 and the original Condition #16 to be Condition 
#18?  What do you think? 
 
The author believes SAS will not add more conditions that contribute to &SYSINFO.  And here are the reasons.  First, 
no condition is more fatal than FATAL ERROR. So new conditions cannot be after Condition #16!  And additional 
conditions could not be added before the current Condition #16 because SAS NEEDS TO BE BACKWARD 

COMPATIBLE!   So the author believes these 16 conditions will remain intact!  And interested SAS programmers can 
do their own programming to uncover any additional discrepancies of their interest! 
 

EXPAND THE USE OF &SYSINFO 

 
The use of &SYSINFO should not be limited to comparing a pair of datasets.  It can be used to validate all analysis 
datasets or reports in a Clinical Study Report (CSR).  In fact, all comparison results can be easily stored in a single 
dataset so that we can know the matches and mismatches at a glance.  Here is how this can be done.   
 

Program 5. Pseudo-code on how result for each program is saved 
 

 

/******************************************************************/ 

/* Program:    v_t1_demog.sas                                     */   

/* Objective:  To demonstate the layout of each validation        */ 

/*             program.                                           */   

/*             Pseudo-Code for Demonstration.                     */ 

/******************************************************************/ 

                                                                                                                

  *----------------------------------------------------------------*; 

*  Initialize work environment.                                  *;                                                               

*  Define libnames.                                              *;                                                                            

*----------------------------------------------------------------*; 

proc datasets lib=work kill nolist nowarn; quit; 

 

libname sdtm '...'; 

libname adam '...'; 

libname vout '...'; 

 

*-------------------------------------------------------------------*; 

*  Step 1                                                           *;                            

*-------------------------------------------------------------------*; 

*  Define macro variables.                                          *;                               

*-------------------------------------------------------------------

%let pgname=v_t1_demog; 

%let tabno=1; 

%let title=Basic Demographic Characteristics; 

%let pop_text=Full Analysis Set; 

%let popfl=FASFL; 

 

%let CUTDATE=2026-02-16; 

*--- Use today() and time() to get the actual execution date and --*; 

*--- time.  Do not use &sysdate and &systime as these macro      --*; 

*--- macro variables give you the time the system started.       --*;         

%let exec_date = %sysfunc(today(), e8601da.); 

%let exec_time = %sysfunc(time(), time5.); 

 

*-------------------------------------------------------------------*; 

*  Step 2                                                           *;                                                                                       

*-------------------------------------------------------------------*; 

*  Create data for comparison.                                      *;                              

*-------------------------------------------------------------------*; 

Program 5. Pseudo-code on how result for each program is saved (continue) 
 

/*    BODY OF THE PROGRAM HERE.                    */ 

 

*-----------------------------------------------------------------*; 

*  Step 3                                                         *;                                                                                       

*-----------------------------------------------------------------*; 

*  Perform PROC COMPARE.                                          *;                                                                       

*-----------------------------------------------------------------*; 

    

*--- Perform PROC COMPARE and get the return codes. ---*;    

PROC COMPARE BASE=BASE DATASET COMPARE=COMPARE DATASET;  

     ID IDVARS;  

run; 

 

*--- Remember to get &SYSINFO right after PROC COMPARE or it will be 

reset. ---*; 

%let RC=&SYSINFO; 

%put &=RC; 

 

%INTERPET(RC=&RC); 

 

*-----------------------------------------------------------------*; 

*  Step 4                                                         *; 

*-----------------------------------------------------------------*; 

*  Store your qc results and related.                             *;                                                                 

*-----------------------------------------------------------------*; 

 

  *--- Add result to QC result dataset. ---*; 

  data qc_&pgname; 

     length TABNO $5 TITLE $100 POPULATION $50 PROGRAM $36 STATUS $4     

            RC 8. RC_1-RC_16 $3 EXTRACT_DATE $15 QC_EXEC_DATE $15   

            QC_EXEC_TIME $15 QC_EXEC_DTM $32 COMMENT $200 COMMENT2          

            COMMENT3 $200; 

  

  TABNO="&TABNO"; 

  TITLE="&TITLE"; 

  Population="&POP_TEXT"; 

  PROGRAM="&pgname"; 

  RC=&RC; 

   

  *--- &RC_1,..., &RC_16 come from invocation of %INTERPRET. ---*; 

  RC_1=&RC_1; RC_2=&RC_2; RC_3=&RC_3;  … RC_16=&RC_16; 

    

  *--- Since these datasets are for the tables, not analysis ---*; 

  *--- dataset, we can allow some differences in attributes. ---*; 

  *--- So &RC < 64 is a PASS here.                           ---*; 

  *--- If this is for analysis dataset, we may need &RC=0, a ---*; 

  *--- perfect match in order to pass.                       ---*; 

  Program 5. Pseudo-code on how result for each program is saved (continue) 
 

  if &RC < 64 then STATUS=’PASS’; 

     else if &RC >= 64 then STATUS=’FAIL’; 

   

  EXTRACT_DATE="&CUTDATE"; 

  QC_EXEC_DATE="&EXEC_DATE"; 

    

  %let len_time=%length(&exec_time);                                                                                                            

  %if &len_time le 4 %then %do;  

      %let exec_time = 0&exec_time;  

  %end;                                                                              

  QC_EXEC_TIME="&EXEC_TIME"; 

  QC_EXEC_DTM=catx('T', QC_EXEC_DATE, QC_EXEC_TIME); 

 

  *--- COMMENT, COMMENT2 and COMMENT3 hold values in the title, ---*;    

 *--- footnote and column headers.                             ---*; 

 *--- These values are not compared in PROC COMPARE. But you   ---*; 

 *--- have to make sure that they are correct.                 ---*;  

  COMMENT="Drug A: N=&NTRT1; Drug B: N=&NTRT2"; 

  COMMENT2="Number of Drug A Subjects from Phase 1: &NPH1_TRT1"; 

  COMMENT3="Number of Drug B Subjects from Phase 1: &NPH1_TRT2 "; 

   

  keep TABNO TITLE POPULATION PROGRAM STATUS RC RC_: QC_EXEC_DTM         

        EXTRACT_DATE COMMENT COMMENT2 COMMENT3; 

  run; 

   

*--- Name the output dataset vout.aaa_qc_results_CSR to easily ---*; 

*--- locate this dataset in the directory because it always    ---*; 

*--- comes first.                                              ---*;  

proc append base=vout.aaa_qc_results_CSR data=qc_&pgname force; 

run; 

 

*--- Keep only the latest result for each program.           ---*; 

*--- Each program result is represented by on observation in ---*; 

*--- vout.aaa_qc_result_CSR.                                 ---*; 

proc sort data=vout.aaa_qc_result_CSR out=vout.aaa_qc_result_CSR; 

   by TABNO QC_EXEC_DTM; 

run; 

 

data vout.aaa_qc_result_CSR; 

  set vout.aaa_qc_result_CSR 

  by TABNO QC_EXEC_DTM; 

  if last.TABNO; 

  run; 

   

*--- End of Program. ---*;   

Program 5 Output:  One observation in vout.aaa_qc_results_csr generated based on the  
                                  pseudo-code that summarizes the result of the comparison.   
                                  (RC_1, …, RC_16 are omitted here due to limited space.) 
 
So each program follows the code like the pseudo-code, one will have one dataset summarizing the result of all the 
reports, like the output below.   
 
 

 
 

 
 
 

Summary Output 1: SAS dataset vout.aaa_qc_result_CSR summarizes the result of the 6 reports. 
 

EXPLORE TRAFFIC-LIGHTING 
 
Based on the result in dataset vout.aaa_qc_result_CSR, we can see that there is one report has a RC=0 indicating a 
Perfect Match, four reports with RC < 64 indicating some discrepancies in attributes and one report with some serious 
problem because it has RC >= 64.   Since the goal here is to validate the datasets used in PROC REPORT, variable 
attributes are not that critical.  Hence, a RC < 64 is considered to be a PASS and a RC >= 64 a FAIL.  If this is a 
comparison of analysis datasets, a perfect match will be required.  As a result, a perfect match (RC=0) is regarded as 
a PASS and any RC > 0 a FAIL.  Now we can put this dataset of comparison status vout.aaa_qc_result_CSR in a 
report and use Traffic-Lighting techniques to highlight the result of the comparison.  The following code can achieve 
this result. 

proc format; 

   value rcfmt 

       0='Green'         /* Perfect Match */ 

       0- <64='Orange'   /* Some discrepancies in Attributes */ 

       64-High='Red'     /* Some critical discrepancies     */ 

       ; 

        

    value $statfmt  

          'PASS'='Green' 

          'FAIL'='Red' 

          ; 

run; 

            

 

title 'Summary of Validation Results'; 

footnote j=l "Last Modified: %sysfunc(today(), date9.) 

%sysfunc(time(), time5.)"; 

options orientation=LANDSCAPE; 

ods listing close; 

ods rtf file="..\validation\output qc_results_csr.rtf"; 

 

proc report data=vout.aaa_qc_result missing center nowd headline  

   headskip split='~' spacing=1 style(report)={outputwidth=100%}  

   style(header)={just=c topmargin=0.02in bottommargin=0.02in}; 

   

   column TABNO TITLE POPULATION PROGRAM EXTRACT_DATE QC_EXEC_DTM RC      

          STATUS COMMENT COMMENT2 COMMENT3; 

    

     define rc/display style(column)=[background=rcfmt.]; 

     define status / display style(column)=[background=$statfmt.]; 

run; 

 

ods rtf close; 

ods listing; 

  
 

Program 6.  Code for Traffic-Lighting to highlight validation results in a report  

Program 6 Output.  Summary Report with Traffic-Lighting to highlight the value of the return code  
                                  (RC) and the PASS/FAIL status.  

 
 

PUT THE VALIDATION RESULTS IN AN EXCEL FILE 
 
With so many variables in the dataset, it can be a bit difficult to read.  Putting the QC result dataset in a Excel file may 
be the solution.  With Excel output, readers are no longer limited to the size of the page and can put in more variables 
data.   
  
There are at least 2 ways one can put the result in an Excel file. 
 
(1).  PROC EXPORT 
(2).  ODS EXCEL 
 
ODS EXCEL will be more ideal since it has nice features that enable us to add style, formats, graphs or the ability to 
add filters, frozen headers, sort data or name the worksheet, etc.  PROC EXPORT does not have that ability. 

Consider the following code to create Excel output.   
 

ods listing off; 

ods EXCEL file="&study\validation\output\qc_results_csr.xlsx";  

ods EXCEL options(sheet_name="QC Result" embedded_titles="yes"  

                  embedded_footnotes="yes"  

                  frozen_headers="yes" frozen_rowheaders="2" 

                  autofilter=’7-8’ hidden_columns=’5,6’ 

                  absolute_column_width='10, 20, 20, 10, 10, 10, 10,  

                   10, 50, 50, 50'); 

 

proc report data=vout.aaa_qc_result_csr missing center nowd headline  

   headskip split='~' spacing=1 style(report)={outputwidth=100%}  

   style(header)={just=c topmargin=0.02in bottommargin=0.02in}; 

   

   column TABNO TITLE POPULATION PROGRAM EXTRACT_DATE QC_EXEC_DTM RC      

          STATUS COMMENT COMMENT2 COMMENT3; 

    

     define rc/display style(column)=[background=rcfmt.]; 

     define status / display style(column)=[background=$statfmt.]; 

run; 

 

ods EXCEL close;  

ods listing;  

 

Program 7.  Code to put the SAS dataset in an Excel file with special features.   
 

 
 
Program 7 Output: Excel Workbook qc_results_csr.xlsx with a sheet named ‘QC Result’. Both 
traffic-lighting and Excel features have been applied 
 
It is very similar to the code in Program 6 but with simple ODS statements, we have converted our output to Excel file 
output with the following Excel features: 
 

• Output has embedded title/footnote. 

• The header row and the first 2 columns are frozen so that the results are more readable.   

• Filtering has been applied to Columns 7 and 8 (Columns for RC and STATUS).   

• Columns 5 and 6 (Columns for EXTRACT_DATE and QC_EXEC_DTM ) have been hidden.   
 
Note that one can use this method to hide the status of the 16 possible conditions, and only to unhide them when 
needed.  One can apply these traffic-lighting techniques and Excel features at the end of each program.  The result 

will be an updated color-coded Excel spreadsheet summarizing the result of all the comparisons. To explore more on 
the features for Excel output, interested readers may want to read the papers by Sekar (2022) and Mahajani (2019).   

CAVEAT #1 

In real life, sometimes values may not be exactly matched.  There can be rounding and computational differences.  
These differences are usually small.  But since PROC COMPARE, by default, uses the EXACT method, any 
discrepancies will not be tolerated.  Users, however, can choose METHOD=RELATIVE, METHOD=PERCENT or 
METHOD=ABSOLUTE and specify the tolerance by CRITERION option.  In such cases, minor discrepancies within 
the tolerance limit are treated as equal.  And Condition 13 (“A value comparison was unequal.”) in Table 1 is not 
violated.  &SYSINFO can still have a zero value but it is a perfect match with some tolerance!  Under the 
circumstances, you may not see our favorite statements in SAS output: 

  

NOTE:  No unequal values were found.  All values compared are exactly equal.   

 

As an example, please consider the following pseudo-code.: 

 

PROC COMPARE BASE=base dataset COMPARE=compare dataset  

             METHOD=RELATIVE CRITERION=0.0001; 

    ID variable-1 variable-2 …; 

run;  

      

Program 8.  Pseudo-code for Comparison using METHOD=RELATIVE and CRITERION=0.0001. 
 

 

 

  Program 8 Output:  A sliver of possible PROC COMPARE output when METHOD=RELATIVE and       
  CRITERION=0.0001 have been applied. 

Note that no matter which METHOD you use and what criterion you choose, as long as the differences in 
values are within the tolerance limit stated, slight discrepancies would not affect how &SYSINFO 
operates!    In other words, with METHOD=RELATIVE and CRITERION=0.0001 then a difference less 
than 0.0001 will be treated as if there are no differences and Condition 13 (A value comparison was 
unequal.) in Table 1, has not been violated!   

CAVEAT #2 

The fact that your program using PROC COMPARE runs without error does not imply the BASE dataset 
and COMPARE dataset run are matching.  SAS, by default, does not produce an ERROR or WARNING 
message if the 2 datasets do not match.  However, since SAS version 9,  ERROR and WARNING options 
have been introduced in PROC COMPARE to require SAS to explicitly specify in the SAS log the ERROR 
and WARNING messages respectively.   

As an example, consider the following PROC COMPARE code using CLASS_PROD and CLASS_QC 
datasets with ERROR and WARNING options, respectively.  The PROC COMPARE result is not affected 
by these options.  However, there will be ERROR or WARNING messages based on the option used.   

 

 
 

Program 9 and Output: The ERROR option in PROC COMPARE produced the red ERROR message 
above. 

 

 
 

Program 10 and Output: The WARNING option in PROC COMPARE produced the green WARNING 
message above in SAS log. 

Note that the messages produced by ERROR and WARNING messages are identical, except ERROR 
messages begin with ‘ERROR: ‘ in red text and WARNING messages begin with ‘WARNING: ‘ in green 
text. 

CAVEAT #3 

Note that an observation will be generated in SAS dataset vout.aaa_qc_result_CSR if the program runs 
successfully.  If the program fails, there will be no observation for that program in 
vout.aaa_qc_result_CSR.  In that case, programmers need to fix the error(s) in the corresponding 
program.   

CONCLUSION  

&SYSINFO from PROC COMPARE holds the value of the return code for each comparison.  It is powerful indicator 
that summarizes the result of the comparison.  Based on this single value, one can tell if it is a perfect match, some 
mismatches in attributes only or some more serious mismatches, such as mismatches in values, variable counts, 
variable types and observation counts.  With the application of %INTERPRET to this return code, users know exactly 
which of the 16 conditions, if any, have been violated. 

By storing the values of &SYSINFO for each comparison in a resulting dataset, one can tell at a glance how all 
comparisons are matching.  Hence, &SYSINFO can help us speed up the validation process.  Moreover, one can use 
traffic-lighting techniques to highlight the result of the comparison.  By converting aforementioned resulting dataset 
into an Excel file, one can take advantage of the Excel features to sort, filter, freeze and hide.  These features will 
make the result more readable.  Armed with this knowledge, one can speed up the validation process without 
compromising its quality!  And Speeding Up Your Validation Process is truly as Easy as 1, 2 and 3!     

APPENDIX:  %INTERPRET 

 

%macro interpret(RC=); 

/****************************************************************************

/* Filename     :  interpret.sas                                            

/* Project      :  Macro Development                                        

/*                                                                          

/* Objective    :  To create a macro named %INTERPRET that                   

/*                                                                          

/*                  (1). Decipher the value stored in &SYSINFO and           

/*                       uncover which of the 16 conditions have been        

/*                       violated.                                           

/*                                                                          

/*                  (2). Create macro varaibles RC_1, ..., RC_16.            

/*                       These macro variables contain a value of Y if the  

/*                       corresponding condition has been violated;          

/*                       otherwise, these macro variable values will be      

/*                       blank.                                              

/*                                                                           

/* Author       :  Alice Cheng                                               

/* Date Created :  2026-03-2                                                 

/*                                                                          

/* Input Data   :  Input &SYSINFO value to the parameter RC of this macro.  

/*                                                                          

/* Output Data  :  Violation condition (if any) will be displayed in the SAS 

/*                 log.                                                     

/*                 Values for Return code of each of the 16 codes, stored in 

/*                 macro variables RC_1, ..., RC_16 will also be displayed in                                                

/*                 the SAS log.                                             

/*           

/***************************************************************************/ 

options nosymbolgen mprint mlogic;                                      

%global RC_1 RC_2 RC_3 RC_4 RC_5 RC_6 RC_7 RC_8 RC_9 RC_10 RC_11 RC_12           

        RC_13 RC_14 RC_15 RC_16;                                          

%let RC_1=; %let RC_2=; %let RC_3=; %let RC_4=;                           

%let RC_5=; %let RC_6=; %let RC_7=; %let RC_8=;                            

%let RC_9=; %let RC_10=; %let RC_11=; %let RC_12=;                        

%let RC_13=; %let RC_14=; %let RC_15=; %let RC_16=;  

data _null_;                                                                    

   /* Perfect match */                                                        

   if &rc=0 then do;                                                        

      put '<<<< Congratulations!  Perfect Match!';  

   end;                                                                      

   else do;                                                                     

   /* Condition 1: Test for data set label */                                

      if &rc = '1'b then do;                                                

         put '<<<< Condition 1: Data sets have different labels';                                                                    

         call symputx('RC_1', 'Y');                                            

      end; 

APPENDIX: %INTERPRET (CONTINUED)        
 
 

/* Condition 2: Test for data set types */                                

      if &rc = '1.'b then do;                                                

         put '<<<< Condition 2: Data set types differ'; 

      call symputx('RC_2', 'Y');                                                   

    end; 

    /* Condition 3: Test for variable informat */                             

    if &rc = '1..'b then do;                                                  

       put '<<<< Condition 3: Variable has different informat';                                                                    

       call symputx('RC_3', 'Y');                                            

    end;                                                                      

    /* Condition 4: Test for variable format */                              

    if &rc = '1...'b then do;                                               

       put '<<<< Condition 4: Variable has different informat';                                                                          

       call symputx('RC_4', 'Y');                                           

     end;                                                                         

    /* Condition 5: Test for length */                                      

    if &rc = '1....'b then do;                                             

        put '<<<< Condition 5: Variable has different lengths';                                                                   

        call symputx ('RC_5', 'Y');                                                    

     end;                                                                        

     /* Condition 6: Test for label */                                        

     if &rc = '1.....'b then do; 

        put '<<<< Condition 6: Variable has different label';                                                             

        call symputx ('RC_6', 'Y');                                                 

     end;                                                                     

    /* Condition 7: Test if base dataset has observation not in compare  

                    dataset */                                                

    if &rc = '1......'b then do;                                              

          put '<<<< Condition 7: Base dataset has observation not in compare 

dataset';                                        

          call symputx ('RC_7', 'Y');                                        

     end;                                                                    

     /* Condition 8: Test if compare dataset has observation not in base  

                     dataset */                                               

     if &rc = '1.......'b then do;                                            

        put '<<<< Condition 8: Compare dataset has observation not in base 

dataset';                                           

        call symputx ('RC_8', 'Y');                                                    

     end;                                                                    

  /* Condition 9: Test if base dataset has BY group not in compare dataset */  

     if &rc = '1........'b then do; 

        put '<<<< Condition 9: Compare dataset has observation not in base 

dataset';                                          

        call symputx ('RC_9', 'Y');                                           

     end;                                                                    

APPENDIX: %INTERPRET (CONTINUED)        
 
 

/* Condition 10: Test if compare dataset has BY group not in base dataset */     

    if &rc = '1.........'b then do;                                          

       put '<<<< Condition 10: Compare dataset has observation not in base dataset';                                           

       call symputx ('RC_10', 'Y');                                             

     end;                                                                     

/* Condition 11: Test if Variable in base data set not in compare data set */      

   if &rc ='1..........'b then do;                                              

      put '<<<< Condition 11: Variable in base data set not found in comparison data 

set';                                           

      call symputx ('RC_11', 'Y');                                            

   end;                                                                      /* 

Condition 12: Test if Comparison data set has variable not in base data   

set */                                                                        

   if &rc = '1...........'b then do;                                         

      put '<<<< Condition 12: Comparison data set has variable not contained in the 

base data set';                               

      call symputx ('RC_12', 'Y');                                                       

   end;                                                                     /* 

Condition 13: Test for values */                                           

   if &rc = '1............'b then do;                                        

      put '<<<< Condition 13: A value comparison was unequal';                                                                   

      call symputx ('RC_13', 'Y');                                            

   end;                                                                     /* 

Condition 14: Test for Conflicting variable types */                            

   if &rc ='1.............'b then do;                                        

      put '<<<< Condition 14: Conflicting variable types between the two data sets 

being compared';                            

      call symputx ('RC_14', 'Y');                                          

   end;                                                                      /* 

Condition 15: Test if BY variable do not match */                          

   if &rc ='1..............'b then do;                                       

      put '<<<< Condition 15: BY variables do not match';                                                                     

      call symputx ('RC_15', 'Y');                                              

   end;                                                                      /* 

Condition 16: Test for Fatal Error: comparison not done */                 

   if &rc ='1...............'b then do;                                      

      put '<<<< Condition 16: Fatal Error: comparison not done';                                                                  

      call symputx ('RC_16', 'Y');                                               

    end;                                                                            

end;                                                                                

run;  

%put &=RC_1  &=RC_2 &=RC_3 &=RC_4;                                          

%put &=RC_5  &=RC_6 &=RC_7 &=RC_8;                                          

%put &=RC_9  &=RC_10 &=RC_11 &=RC_12;                                              

%put &=RC_13  &=RC_14 &=RC_15 &=RC_16; 

%mend interpret; 

 

 

 


